

Flight Plan Route

30 November, 2001

Cristian Cadar and Tudor Leu

The flight plan route for a flight is received by CTAS half an hour to several hours before the first track information arrives. The route of a flight plan is initialized with an ADD_FLIGHT_PLAN message and can be changed at any time before the last track data (SENDING_AIRCRAFT message) arrives with an AMEND_FLIGHT_PLAN message.

The flight plan route is not meant to be a complete description of the path the plane is flying. Instead, it describes only the *horizontal intent*.

A route contains two types of elements:

- fixes
- predefined routes

Fixes can be specified in the following 5 manners:

- **Predefined fix:** a sequence of 3 capital letters. Ex: TXK, LIT, BOS. In the grammar we construct we refer to this type of fix as <Predef_fix>.
- **Fix specified by its geographical coordinates:** a sequence of 4 digits specifying the latitude of the fix (optionally followed by the letter N), followed by a slash, followed by a sequence of 5 digits specifying the longitude of the fix (optionally followed by the letter W): nnnn(N)/nnnnn(W). Ex: 3441N/10332W, 3441/10332. In the grammar we construct we refer to this type of fix as <LatLong_fix>.
- **Radial fix:** radial fixes are specified relative to another predefined fix. The general format is <lll predef_fix><nnn radial from the predefined fix><mmm miles away> Ex: BOS263039. In the grammar we construct we refer to this type of fix as <Radial_fix>.
- **Coordinate fix:** coordinate fixes are specified by their absolute coordinates in the Center: <nnn x_coordinate><nnn y_coordinate>. Ex: 234/678.
- **Intersection fix:** these fixes are situated at the intersection of two jet-routes with no nav aid. Ex: CUGAR

In addition to fixes, routes can contain predefined routes, which describes commonly-used routes in high traffic. There are 3 types of predefined routes:

- **SID routes:** SID stands for Standard Instrument Departure and represents a sequences of actions and paths to be taken in the transition between terminal area and en-route airspace. A SID route is represented by a sequence of 3 letters followed by a single digit. Ex: SID7, GDM2. In the grammar we construct we refer to this type of route as <SID_route>.

- **STAR routes:** STAR stands for Standard Terminal Arrivals and represents a sequences of actions and paths to be taken in the transition between en-route and terminal airspace. A STAR route is represented by a sequence of 4 letters followed by a single digit. Ex: STAR2, DALL5. In the grammar we construct we refer to this type of route as <STAR_route>.
- **Jet-Routes:** These are predefined-routes at high altitudes (18,000 – 45,000 ft). In the grammar we construct we refer to this type of route as <Jet_route>.
- **Victor-Routes:** These are predefined-routes at low altitudes (1,200 – 17,999 ft). In the grammar we construct we refer to this type of route as <Victor_route>.

Usually, a flight plan route connecting two major airports is composed of:

- a SID route
- an en-route path
- a STAR route

However, in the unlike case in which the traffic at the departure or arrival airport is not very high, the SID or STAR routes can be omitted.

An example of a standard flight-plan route between Dallas Fort Worth Airport and Boston Logan Airport is:

DFW./LAS.DALL5.TXK..LIT.J131.PXV.J29.JHW.V82.ALB.GDM2.BOS

Note that each predefined route is surrounded by two predefined fixes representing the start and the end fixes of the respective route. For example the jet-route J131 starts with the fix LIT and ends with the fix PXV.

Also, there are 3 types of separators:

- **".."** which separates two fixes, specified in any form. Ex: TXK..LIT, 3245N/10234W..DFW, IRW..IRW263039.
- **."** which separates predefined routes by the predefined fixes surrounding them. Ex: LAS.DALL5.TXK, LIT.J131.PXV, JHW.V82.ALB.
- **"/."** which indicates that part of the route was left out because the plane already passed them

The grammar we defined is the following:

- **Fixes**

```
<Fix> := <Predef_fix> | <LatLong_fix> | <Radial_fix> |  
        <Coordinate_fix> | <Intersection_fix>  
  
<Predef_fix> := <letter> <letter> <letter>  
  
<LatLong_fix> := <Lat> "N" "/" <Long> "W" | <Lat> "/" <Long>  
<Lat> := <digit> <digit> <digit> <digit>  
<Long> := <digit> <digit> <digit> <digit> <digit>  
  
<Radial_fix> := <Predef_fix> <Radial><Distance>  
<Radial> := <digit> <digit> <digit>  
<Distance> := <digit> <digit> <digit>  
  
<Coordinate_fix> := <Coord> "\" <Coord>  
<Coord> := <digit> <digit> <digit>  
  
<Intersection_fix> := <letter> <letter> <letter> <letter>  
<letter>
```

- **Predefined Routes**

```
<Predef_route> := <SID_route> | <STAR_route> | <Jet_route> |  
                <Victor_route>  
  
<SID_route> := <letter> <letter> <letter> <digit>  
  
<STAR_route> := <letter> <letter> <letter> <letter> <digit>  
  
<Jet_route> := "J" <Number>  
  
<Victor_route> := "V" <Number>  
  
<Number> := <digit> | <digit> <Number>
```

- **Separators**

```
<Separator1> := ".."  
<Separator2> := "."  
<Separator3> := "./."
```

- **Route**

```
<Route> := <Fix> | <Fix> <From_fix>  
  
<From_fix> := <Separator1> <Route> |  
             <Separator2> <Predef_route> <Separator2> <Route> |  
             <Separator3> <Route>
```